

Rock, Scissors, Paper

Remember the game of rock-scissors-paper. You and your opponent make a motion three times with your fists and then either show a flat hand (paper), fist (rock), or two fingers (scissors). Depending upon what is shown, the game is a tie (both show the same) or one person wins. Paper wraps up rock, so it wins. Scissors cut paper, so they win. And rock breaks scissors, so it wins.

In this computerized version of rock-scissors-paper, you can play up to ten games vs. the computer.

Charles Lund wrote this game while at the American School in The Hague, Netherlands.

GAME OF ROCK, SCISSORS, PAPER
CREATIVE COMPUTING MORRISTOWN, NEW JERSEY

```

10 PRINT TAB(21);"GAME OF ROCK, SCISSORS, PAPER"
20 PRINT TAB(15);"CREATIVE COMPUTING MORRISTOWN, NEW JERSEY"
25 PRINT:PRINT:PRINT
30 INPUT "HOW MANY GAMES";Q
40 IF Q<11 THEN 60
50 PRINT "SORRY, BUT WE AREN'T ALLOWED TO PLAY THAT MANY.":GOTO 30
60 FOR G=1 TO Q
70 PRINT:PRINT "GAME NUMBER";G
80 X=INT(RND(1)*3+1)
90 PRINT "3=ROCK...2=SCISSORS...1=PAPER"
100 INPUT "1...2...3...WHAT'S YOUR CHOICE";K
110 IF (K-1)*(K-2)*(K-3)<>0 THEN PRINT "INVALID.":GOTO 90
120 PRINT "THIS IS MY CHOICE..."
130 ON X GOTO 140,150,160
140 PRINT "...PAPER":GOTO 170
150 PRINT "...SCISSORS":GOTO 170
160 PRINT "...ROCK"
170 IF X=K THEN 250
180 IF X>K THEN 230
190 IF X=1 THEN 210
200 PRINT "YOU WIN!!!":H=H+1:GOTO 260
210 IF K<>3 THEN 200
220 PRINT "WOW! I WIN!!!":C=C+1:GOTO 260
230 IF K<>1 OR X<>3 THEN 220
240 GOTO 200
250 PRINT "TIE GAME. NO WINNER."
260 NEXT G
270 PRINT:PRINT "HERE IS THE FINAL GAME SCORE:"
280 PRINT "I HAVE WON";C;"GAME(S).
290 PRINT "YOU HAVE WON";H;"GAME(S).
300 PRINT "AND";Q-(C+H);"GAME(S) ENDED IN A TIE."
310 PRINT:PRINT "THANKS FOR PLAYING!!"
320 END

```

HOW MANY GAMES? 10

```

GAME NUMBER 1
3=ROCK...2=SCISSORS...1=PAPER
1...2...3...WHAT'S YOUR CHOICE? 3
THIS IS MY CHOICE...
...PAPER
WOW! I WIN!!!

```

```

GAME NUMBER 2
3=ROCK...2=SCISSORS...1=PAPER
1...2...3...WHAT'S YOUR CHOICE? 2
THIS IS MY CHOICE...
...ROCK
WOW! I WIN!!!

```

```

GAME NUMBER 3
3=ROCK...2=SCISSORS...1=PAPER
1...2...3...WHAT'S YOUR CHOICE? 2
THIS IS MY CHOICE...
...PAPER
YOU WIN!!!

```

```

GAME NUMBER 4
3=ROCK...2=SCISSORS...1=PAPER
1...2...3...WHAT'S YOUR CHOICE? 1
THIS IS MY CHOICE...
...ROCK
YOU WIN!!!

```

```

GAME NUMBER 5
3=ROCK...2=SCISSORS...1=PAPER
1...2...3...WHAT'S YOUR CHOICE? 3
THIS IS MY CHOICE...
...SCISSORS
YOU WIN!!!

```

```

GAME NUMBER 6
3=ROCK...2=SCISSORS...1=PAPER
1...2...3...WHAT'S YOUR CHOICE? 2
THIS IS MY CHOICE...
...SCISSORS
TIE GAME. NO WINNER.

```

```

GAME NUMBER 7
3=ROCK...2=SCISSORS...1=PAPER
1...2...3...WHAT'S YOUR CHOICE? 2
THIS IS MY CHOICE...
...ROCK
WOW! I WIN!!!

```

```

GAME NUMBER 8
3=ROCK...2=SCISSORS...1=PAPER
1...2...3...WHAT'S YOUR CHOICE? 3
THIS IS MY CHOICE...
...ROCK
TIE GAME. NO WINNER.

```

```

GAME NUMBER 9
3=ROCK...2=SCISSORS...1=PAPER
1...2...3...WHAT'S YOUR CHOICE? 1
THIS IS MY CHOICE...
...ROCK
YOU WIN!!!

```

```

GAME NUMBER 10
3=ROCK...2=SCISSORS...1=PAPER
1...2...3...WHAT'S YOUR CHOICE? 2
THIS IS MY CHOICE...
...ROCK
WOW! I WIN!!!

```

HERE IS THE FINAL GAME SCORE:
I HAVE WON 4 GAME(S).
YOU HAVE WON 4 GAME(S).
AND 2 GAME(S) ENDED IN A TIE.